

Basis-Auswertung der Lebenslagenbefragungen mit SAS – Eine Architektur der Module

Anja Nimmergut
Statistisches Bundesamt
Gustav-Stresemann-Ring 11
65189 Wiesbaden
anja.nimmergut@destatis.de

Tatjana Theis
Statistisches Bundesamt
Gustav-Stresemann-Ring 11
65189 Wiesbaden
tatjana.theis@destatis.de

Zusammenfassung

Die Bundesregierung hat das Statistische Bundesamt in ihrem Arbeitsprogramm „Bessere Rechtsetzung 2014“ damit beauftragt, eine Befragung über die Wahrnehmung von behördlichen Dienstleistungen zu entwickeln und durchzuführen. Sowohl Bürgerinnen und Bürger als auch Unternehmen werden dabei nach ihren subjektiven Erfahrungen mit Behördenkontakten bei bedeutenden oder häufigen Ereignissen gefragt. Grundlage ist das sogenannte Lebenslagen-Modell. Beispiele der insgesamt 22 Lebenslagen der Bürgerbefragung sind Heirat oder Steuererklärung. Die Unternehmensbefragung beinhaltet insgesamt zehn Lebenslagen, beispielsweise die Gründung eines Unternehmens oder den Bau einer Betriebsstätte. Auf der Internetseite www.amtlich-einfach.de werden Informationen und Ergebnisse zu den Befragungen dargestellt. Darunter findet man auch die sogenannten Behördenreisen, die grafisch alle Behördenkontakte zeigen, die im Rahmen einer Lebenslage anfallen können. Die Ergebnisse der ersten Befragungswelle 2015 wurden auf Pressekonferenzen in 2015 (Bürgerbefragung) und 2016 (Unternehmensbefragung) (Statistisches Bundesamt, 2015 a-c [8]-[10], Statistisches Bundesamt, 2016 a-c [5]-[7]), Konferenzen (Kühnhenrich und Michalik, 2016 [2]) und Publikationen (Himmelsbach, et al., 2016 [3]) vorgestellt. Die zweite Befragungswelle 2017 wird momentan für die Veröffentlichung vorbereitet (geplant April 2018).

Dieser Beitrag erlaubt erstmals einen Einblick hinter die Kulissen: in die SAS-Auswertungsarchitektur, aus der heraus die Ergebnisse generiert werden. Sie beinhaltet eine horizontale „Kennzahlenkomponente“ als auch eine vertikale „Ebenenkomponente“ über die verschiedenen, hierarchischen Aggregationsebenen „individuelles Interview“, „Behörde“, „Schicht“ und „Lebenslage“. Als Abschluss der mehrstufigen Analyse wird ein Mittelwert über alle Lebenslagen als Maßzahl für die Zufriedenheit mit der öffentlichen Verwaltung gebildet. Neben den einzelnen SAS-Auswertungsmodulen mit ihren spezifischen Makros zur Ergebniserstellung entlang der einzelnen Ebenen und Faktoren werden ebenfalls die SAS- Rohdatenplausibilisierung und -Transponierung (Makroeinsatz mit einer „%do“- , „%let“- , „%scan“-Anweisung versus die Prozedur „proc transpose“), die Designgewichtung, Inzidenzberechnung und die grafische Darstellung der Ergebnisse mittels ODS Graphics vorgestellt. Hierzu werden aus den Modulen heraus SAS-Code-Abschnitte oder Makros exemplarisch herausgegriffen, die der Ergebnisermittlung zugrunde liegen, bestehende Aggregationsebenen- oder Datenvolumenprobleme pragmatisch lösen bzw. die grafische Ergebnisausgabe steuern.

Schlüsselwörter: Lebenslagen, Datenplausibilisierung, Designgewichtung, Inzidenz, ODS Graphics

1 Ziele der Auswertungsstrategie und konzeptioneller Rahmen

Die Lebenslagenbefragung soll ermitteln, wie Bürgerinnen/ Bürger und Unternehmen innerhalb bestimmter Lebenslagen den Kontakt und die Zusammenarbeit mit der Verwaltung subjektiv wahrnehmen. Aus den Ergebnissen werden auf Regierungsebene Vereinfachungs- und Optimierungsmaßnahmen erarbeitet. Die Befragung wird wiederholt, wobei die Befragungswellen untereinander vergleichbar sein sollen, aber auch weiterentwickelt werden müssen.

Konzeptionell basiert die Umsetzung auf dem Lebenslagen-Modell. Es untersucht ausgewählte, zentrale Lebenslagen und Behörden (Schmidt, et al., 2015 [4]). Die Konzeption der Befragung ist nicht Gegenstand dieses Papiers. Die grundlegende Architektur wird nur insofern beschrieben, als sie für das Verständnis der Auswertungen notwendig ist.

Das vorliegende Papier beschreibt die Auswertungsstrategie, die verfolgt wird, um die oben genannten Ziele zu erreichen. Die über die Befragung gewonnen Rohdaten (Himmelsbach, et al., 2016 [3]) müssen zunächst einmal Plausibilitätsprüfungen unterzogen werden. Elementare Voraussetzung der Ergebniserstellung ist gleichfalls die Berechnung eines Designgewichts als auch die Ermittlung der vor der Datenerhebung unbekanntes „Häufigkeiten“ bzw. dem „Vorkommen“ von Lebenslagen, Schichten und Behördenkontakten, auch Inzidenzen genannt.

Das Ziel „Ermittlung der Wahrnehmung von Kontakt und Zusammenarbeit mit der öffentlichen Verwaltung“ wird über eine Basisauswertung der erhobenen Daten erreicht: Ergebnis sind Zufriedenheitswerte in aggregierter Form. Aufbauend werden Teilmengen- und Streuungsauswertungen durchgeführt, um erste Hinweise zu erhalten, wo Vereinfachungs- und Optimierungsmaßnahmen am ergiebigsten sein könnten. Alle Ergebnisse werden mittels SAS/GRAPH und ODS Graphics als Grafik für die Veröffentlichung zur Verfügung gestellt

Jeder dieser Auswertungsabschnitte (1. Plausibilisierung – 2. Designgewichtung/ Inzidenzberechnung – 3. Basisauswertung – 4. Teilmengen-/ Streuungsauswertungen – 5. Grafische Darstellung) beinhaltet aufwändige, aufeinander aufbauende Konzeptions- und Programmierarbeiten. Sie müssen in einem (auch zeitlich) gestuften Verfahren geplant und durchgeführt werden.

2 Architektur des Lebenslagen-Modells und seine Erfordernisse an die Auswertung

Die Zufriedenheit der Bürger und Unternehmen mit behördlichen Dienstleistungen wird mit Hilfe eines Kennzahlensystems ermittelt, welches insgesamt 16 Einzelfaktoren enthält. Diese sind z. B. „Verständlichkeit der Formulare und Anträge“, „Öffnungszeiten“, „Gesamte Verfahrensdauer“. Die Kennzahlen werden für die zuvor festgelegten Behörden, mit denen eine Person oder ein Unternehmen innerhalb verschiedener Lebenslagen, z. B. „Kinderbetreuung“, „Einstellen von Beschäftigten“, in Interaktion tritt auf einer fünfstufigen Zufriedenheitsskala unter Angabe der jeweiligen Wichtigkeit der Kennzahl erfragt (Himmelsbach, et al., 2016 [3]).

Um den Kreis der befragten Behörden zu erweitern, wird den Befragten zusätzlich die Möglichkeit gegeben zu weiteren Behörden eine „Kurzauskunft“ ohne Kennzahlabfrage zu geben. Die Zufriedenheit mit diesen sogenannten „flach“ befragten Behörden wird direkt, d. h. ohne die detaillierte Abfrage der Kennzahlen und deren Wichtigkeit, angegeben.

Die Personen bzw. Unternehmen werden verschiedenen vorab definierten Schichten einer Lebenslage zugeordnet (z. B. „Krippe“, „Kindergarten“, „Hort“, „sonstige Betreuungsangebote“ in der Lebenslage „Kinderbetreuung“), um eine ausreichende Fallzahl von unterschiedlichen Bürger- oder Unternehmenspopulationen im Datensatz vertreten zu haben.




Abbildung 1: Auswertungskomponenten der Lebenslagenbefragung

Die Auswertung des Lebenslagen-Modells beinhaltet somit eine horizontale „Kennzahlenkomponente“ (ohne flach befragte Behörden), als auch eine vertikale „Ebenenkomponente“ über die verschiedenen Aggregationsebenen: (1.) individuelles Interview einer Person/ eines Unternehmens, (2.) für eine Behörde, (3.) innerhalb einer Schicht und (4.) für eine Lebenslage. Schließlich wird (5.) ein Mittelwert über alle Lebenslagen als Maßzahl für die Zufriedenheit mit der öffentlichen Verwaltung berechnet. Abbildung 1 gibt einen Überblick über die verschiedenen Ebenen.

Für jede Befragungswelle muss zuerst eine systematische Basisauswertung durchgeführt werden, die die Einzeldaten entlang dieser vertikalen und horizontalen Komponenten aggregiert. Wie in Abbildung 1 dargestellt, ist die Zufriedenheit mit einer Behörde innerhalb einer Befragenschicht, die einer Lebenslage zugeordnet ist, die unterste Auswertungsstufe. Aufbauend auf diesem systematisch-aggregierten Überblick können Teilmengen- und Streuungsauswertungen (z. B. eine Betrachtung der unzufriedensten Befragungsteilnehmer oder eine Auswertung nach Wirtschaftszweigen) durchgeführt werden. Sie benötigen die zuvor geschaffene Auswertungsarchitektur der Basisauswertungen plus spezifische individuelle Filtersetzungen, die in die Basisauswertungen eingepasst werden müssen.

3 Erste Auswertungsvoraussetzung: Rohdatentransformation und Datenplausibilisierung

Aus jeder bisher durchgeführten Befragungswelle bekommt das Statistische Bundesamt zwei sogenannte Bruttodateien (eine für die Bürger- und eine für die Unternehmensbefragung) und zwei sogenannte Ämterdateien (jeweils auch eine für die Bürger- und eine für die Unternehmensbefragung) übermittelt. Diese Dateien enthalten alle über die Befragung gewonnen Rohdaten.

Während die Ämterdateien die durchgeführten Nettointerviews zu den einzelnen Behörden beinhalten (d. h. hier werden die eigentlichen Behördenbeurteilungen abgegeben), sind in der Bruttodatei die sogenannten Kontaktinterviews abgelegt. Die Kontaktinterviews dienen der Übersicht darüber, bei welcher Lebenslage die Zielpersonen/ das Unternehmen in den letzten zwei Jahren Kontakt zu einer Behörde oder einem Amt hatten und in welche Schicht sie bzgl. dieser Lebenslage einzusortieren sind.

Nicht zu jedem in der Bruttodatei festgehaltenen Behördenkontakt wurde ein Nettointerview geführt. Ein Limit von maximal sechs Behördenbeurteilungen je Person dient der Beschränkung der Befragtenbelastung. Zudem kann nach einem/wenigen Behördenkontakt-Interview(s) abgekürzt werden, wenn die Interviewdauer für die Zielperson zu lang wurde (Himmelsbach, et al., 2016 [3]).

Während in der Bruttodatei pro Person/ Unternehmen in einer Datensatzzeile alle Informationen zu den Lebenslagen, Schichten und Behörden (diese sind hier als jeweils einzelne Variable angelegt) als 0/1-Kodierung übermittelt wird, wird in der Ämterdatei das Behördeninterview pro Behörde, Schicht und Lebenslage übermittelt. D. h. die Datensätze müssen nach einer einfachen grundsätzlichen Plausibilitätsprüfung zunächst einmal in ein einheitliches Format überführt werden.

Da die Zufriedenheit mit einer Behörde innerhalb einer Schicht, die einer Lebenslage zugeordnet ist, die unterste Auswertungsstufe darstellt (siehe oben) musste die Bruttodatei entsprechend angepasst werden. Alle weiteren notwendigen Auszählungen, sei es die Anzahl der Lebenslageninterviews oder Schichtbesetzungen sind nach dieser Umstellung vergleichsweise einfach, schnell und unkompliziert durchzuführen.

Die Umstellung der Bruttodatei auf Behördendatensätze wurde auf zwei verschiedene Weisen getestet. Sowohl eine Indizierung der Lebenslagen-, Schicht- und Behördenvariablen über ein Makro mit einer „%do“- , „%let“- und „%scan“-Anweisung und anschließendem Zusammenspiel der Einzeldateien, als auch der Einsatz der Prozedur „proc transpose“ ist möglich.

In den Codes hat sich der Einsatz der Prozedur „proc transpose“ durchgesetzt. Dadurch wird eine Reduktion der zu verarbeitenden Einzeldateien erreicht und alle Datensätze, die hinsichtlich der Lebenslagen-, Schicht- und Behördenvariablen ein Missing aufweisen, können in den transponierten Files gelöscht werden. Somit erscheint diese Methode als die effizienteste (vgl. auch Hochgürtel und Lösch, 2013 [1]).

Die Behördenkontakte zu den einzelnen Lebenslagen sind in der Rohdatenbruttodatei in den Variablen „amtl101“ – „amtl130“ angegeben. D. h. bis zu 30 Behördenkontakte konnten angegeben sein. In diesen Variablen ist zunächst zweistellig der Behördenkontakt und danach zweistellig die Lebenslage angegeben, sodass eine vierstellige Kodie-

ung resultiert. Während der Umstellung in eine vertikale Datei mit den Angaben eines Behördenkontakts pro Schicht und Lebenslage mussten die vierstelligen Behördenkontakt-Lebenslagenkodierungen in jeweils eine Variable für die Lebenslage und eine Variable des entsprechenden Behördenkontakts umkodiert werden. Dies wird erreicht, indem die Variablen über eine „put“-Anweisung in eine alphanumerische Variable geändert, die jeweiligen Lebenslagen und Ämterangaben über eine „substr“-Anweisung extrahiert und diese abschließend über eine „input“-Anweisung wieder in numerische Variablen geändert werden.

```
/*Behörden werden für die Bruttodatei transformiert*/

data behoerdentrasform (keep=pagnr amt1101 - amt1130);
set Ordner.&FileBrutto.;
run;

proc transpose data=behoerdentrasform out=behoerdentrasform1;
  by pagnr;
run;

data behoerdentrasform1 (drop= _NAME_ COL LL Amt);
set behoerdentrasform1;
  if COL1 ne .;
  COL = put(COL1,4.);
  LL = substr(COL,3,2);
  Amt = substr(COL,1,2);
  q22 = input(LL,2.);
  q23 = input(Amt,2.);
run;

/*Lebenslagen werden für die Bruttodatei transformiert*/
... /*analog der Behörden*/

/*Schichten werden für die Bruttodatei transformiert*/
... /*analog der Behörden*/
```

Nach der Umstellung werden in den Ämterfiles die Zufriedenheits- und Wichtigkeitsangaben über eine Indizierung mit Hilfe von Arrays die Zufriedenheitswerte umkodiert. Die Einschätzung des Behördenkontakts wurde anhand einer fünfstufigen Likert-Skala erfragt und entsprechend von 1 bis 5 kodiert. Für die Ergebniserstellung wurden sie in Zahlen von -2 (Ablehnung) bis +2 (Zustimmung) umkodiert (Himmelsbach, et al., 2016 [3]).

```
/*Angaben mit "Keine Angabe/trifft nicht zu" werden zu missing umgewandelt */
array alles{*} K1-K16;
do i = 1 to dim(alles);
  if alles{i} in (6,9) then alles{i} = .;
end;
drop i;
```

```
/*Die Skaleneinteilung der einzelnen Kennzahlen reicht in den Rohdaten von "1"(sehr zufrieden) bis "5" (sehr unzufrieden). Insgesamt sollen die Wertigkeiten jedoch von "+2" (sehr zufrieden) über "0" (neutral) bis "-2" (sehr unzufrieden) reichen. Daher werden die gelieferten Wertigkeiten der Kennzahlen umcodiert*/
```

```
array zufrieden{19} K1-K13 K16;
```

```
/*Der Array läuft nicht über alle 16 Kennzahlen, da K14 und K15 gesondert umcodiert werden*/
```

```
do i = 1 to 14;  
    zufrieden{i} = -zufrieden{i} + 3;  
end;
```

```
/*Zufriedenheitsskala für den Eindruck von Diskriminierung und Korruption wird ebenfalls in +2/-2 ungewandelt, von "1"(Ich stimme voll zu) bis "5" (Ich stimme gar nicht zu)*/
```

```
K14 = K14 - 3;  
K15 = K15 - 3;
```

4 Zweite Auswertungsvoraussetzung: Designgewicht- und Inzidenzberechnung

Da die Befragungen als Stichprobenerhebungen über computergestützte Telefoninterviews (CATI) durchgeführt werden (Himmelsbach, et al., 2016), müssen Korrekturen für unterschiedliche Auswahlwahrscheinlichkeiten der befragungsbereiten Personen, Unternehmen und der befragten Behörden erfolgen.

Das hierfür berechnete Designgewicht berücksichtigt die Anzahl von Festnetznummern und Mobilfunkgeräten sowie die Anzahl der Haushaltsmitglieder für die Bürgerbefragung. Es korrigiert eine überproportionale Ziehung von großen Unternehmen, die notwendig ist, damit genügend dieser großen Unternehmen für Auswertungen zur Verfügung stehen. Es berücksichtigt die Einhaltung genauer Schichtvorgaben sowie die Auswahl von Behördenkontakten, wenn eine Person oder ein Unternehmen mit so vielen Behörden Kontakt hatte, dass dem Befragungsrahmen im Sinne einer vertretbaren Befragtenbelastung ein Limit gesetzt werden musste.

Das Designgewicht DG_{Gesamt} der Bürgerdatensätze wird aus der Bruttodatei unter Einbeziehung des Gewichtungsfaktors nach Erreichbarkeit ($DG1_i$), des Gewichtungsfaktors zur Steuerung für die Unterstichprobe innerhalb einer Schicht ($DG2_{\text{Schicht}}$) und des Gewichtungsfaktors für Behörden innerhalb einer Lebenslage ($DG3_{\text{Behörde}}$) berechnet nach:

$$DG_{\text{Gesamt}} = DG1_i * DG2_{\text{Schicht}} * DG3_{\text{Behörde}}$$

Der Gewichtungsfaktor nach Erreichbarkeit ($DG1_i$) wird wie folgt berechnet:

$$DG1_i = \frac{1}{\left(\text{Mobilnummern}_i + \frac{\text{Festnetzanschlüsse}_i}{\text{HH-Größe}_i} \right)}$$

Der Gewichtungsfaktor zur Steuerung für die Unterstichprobe innerhalb einer Schicht ($DG2_{\text{Schicht}}$) berechnet sich nach:

$$DG2_{\text{Schicht}} = \frac{\sum_{i \in S} DG1_i \text{Brutto}}{\sum_{i \in S} DG1_i \text{Netto}}$$

wobei für eine Schicht S über alle Personen summiert wird, die eine Lebenslage der Schicht haben (mit bzw. ohne Interviews). D. h. gegebenenfalls zählt eine Person zu verschiedenen Schichten, wenn sie von mehreren Lebenslagen betroffen ist.

Der Gewichtungsfaktor für Behörden innerhalb einer Lebenslage ($DG3_{\text{Behörde}}$) wird für eine bestimmte Person i in der Lebenslage l berechnet:

$$DG3_{\text{Behörde}} = M_{il}/m_{il},$$

wobei

M_{il} : Zahl der Behörden, mit denen die Person i in der Lebenslage l zu tun hatte,

m_{il} : Zahl der realisierten Interviews der Person i zur Lebenslage l.

Den Faktor erhalten alle realisierten Interviews der Person i und der Lebenslage l.

Das Designgewicht DG_{Gesamt} der Unternehmensdatensätze wird im Prinzip analog berechnet mit dem Unterschied, dass $DG1_i$ der Gewichtungsfaktor nach Wirtschaftszweig (WZ) und Beschäftigtengrößenklasse (BGK) ist.

Dieser wird wie folgt berechnet:

$$DG1_i = \frac{\text{Anteil der Unternehmen pro WZ und BGK in der Grundgesamtheit}}{\text{Anteil der Unternehmen pro WZ und BGK mit Interviews}}$$

Bei der Ermittlung beider $DG1_i$ stellten die Item-Angaben „keine Angabe“ bzgl. Anzahl der Haushaltsmitglieder, der Festnetznummern, der Mobilfunkgeräten (Bürgerbefragung) sowie beim Wirtschaftszweig und der Beschäftigtengrößenklasse die größte Herausforderung dar, denn ohne Werte kann für die einzelnen Behördeninterviews keine Designgewichtung erfolgen.

In der Bürgerbefragung wurden die Lücken über eine einfache Mittelwertbildung aller jeweiligen Datensätze mit Haushaltsmitglieder-, Festnetz- oder Mobilfunkgeräteangabe mit „proc means“ erreicht. Für die Unternehmensdaten wurde die „proc means“-Anweisung in ein Makro eingebettet, dass mit einer „%do“- , „%let“- und „%scan“-Anweisung jeweils beschäftigtengrößenklassenspezifische Mittelwerte für die jeweiligen Branchen berechnet.

/* Für die Interviews mit sonstigem Wirtschaftszweig (id_hilf: 1_19,2_19,3_19,4_19,5_19,6_19) wird der Gewichtungsfaktor nach Wirtschaftszweig und Beschäftigtengrößenklasse $DG1_i$ über den (mit den

A. Nimmergut, T. Theis

Anteilen der Grundgesamtheit) gewichteten Mittelwert der anderen DGli berechnet */

```
%macro BGKWZSonstige ;
    %do t=1 %to 6;
        %let bgnr = %scan(1 2 3 4 5 6 ,&t.);

data BGKWZ_Gewicht&bgnr.;
set BGKWZ_Gewicht;
if id_hilf in ("%bgnr._5" "%bgnr._17" "%bgnr._6" "%bgnr._15"
"%bgnr._16" "%bgnr._3" "%bgnr._1" "%bgnr._11" "%bgnr._7" "%bgnr._10"
"%bgnr._14" "%bgnr._4" "%bgnr._2" "%bgnr._9" "%bgnr._13" "%bgnr._8"
"%bgnr._12" "%bgnr._18");
run;

proc means data = BGKWZ_Gewicht&bgnr. mean noprint;
    var DGli;
    weight Anteil_GG;
    output out = BGK2&bgnr.;
run;

data BGK2&bgnr. (drop = _STAT_ _TYPE_ _FREQ_);
set BGK2&bgnr.;
    if _STAT_ = "MEAN";
    id_hilf = "%bgnr._19";
run;

proc delete data = BGKWZ_Gewicht&bgnr.;
run;

%end;
%mend;
%BGKWZSonstige ;

data BGK2_alle;
set BGK21 BGK22 BGK23 BGK24 BGK25 BGK26;
run;
```

Unter Inzidenz wird die vor der Datenerhebung unbekannte „Häufigkeit“ bzw. das „Vorkommen“ von Lebenslagen, Schichten und Behördenkontakten verstanden. Da sie allesamt unbekannt sind, werden aus dem Bruttodatenatz die Behördenkontakt-, die Schicht- und darauf aufbauend die Lebenslageninzidenzen berechnet. Die Inzidenzen werden benötigt, um das Vorkommen der Schichten innerhalb der Lebenslagen und die Bedeutung der Interaktionen mit den verschiedenen Behörden (Behördenkontakte) innerhalb der Schichten abbilden zu können. Die Einbindung der Designgewichte und Inzidenzen in die Auswertungsebenen wird in Abbildung 1 (siehe oben) verdeutlicht.

5 Vertikale und horizontale Basisauswertung: Vom Interview zur „Zufriedenheit mit der öffentlichen Verwaltung“

Ergebnis der Basisauswertung sind durchschnittliche Zufriedenheiten für alle Auswertungsebenen (vertikale Auswertung) sowie Zufriedenheitswerte für alle 16 Kennzahlen (horizontale Auswertung).

Die als erstes durchgeführte **vertikale Auswertung** berechnet zunächst die Zufriedenheit mit der Behörde: den **Behördenscore**¹. Für jede der 16 Kennzahlen wird für jede Behörde jeder Schicht jeder Lebenslage die durchschnittliche Zufriedenheit gewichtet mit der Wichtigkeit und dem Designgewicht berechnet und zu einem Behördenergebnis aufsummiert. Für die flach befragten Behörden wird direkt aus den kurzen Behördeneinschätzungen ein designgewichteter Behördenscore gebildet. Dieser Auswertungsschritt – die Bestimmung des Behördenscores für alle Schichten aller Lebenslagen – dient als Basis aller weiteren Auswertungen.

```
/*Um von der Ebene der Einzelinterviews zur nächsten Ebene (Amt pro
Schicht pro Lebenslage) zu kommen, werden für die detailliert be-
fragten Behörden die arithm. Mittel der Einzelangaben zu den 16
Kennzahlen pro Amt pro Schicht pro Lebenslage gebildet. Dadurch
aggregieren sich die Einzelangaben - zunächst noch ungewichtet mit
der Wichtigkeit - auf die jeweiligen Ämter */
```

```
proc tabulate data= E4&Daten._Kennzahlen_detail
  out= K_Amt_Schicht_detail (drop= _type_ _page_ _table_);
  var K1-K16 /*Interviewwerte der einzelnen Kennzahlen*/;
  class q22 /*LL*/ q32 /*Schicht*/ q23 /*Behörde*/ / missing;
  table q22*q32*q23,
    N K1*mean K2*mean K3*mean K4*mean K5*mean K6*mean K7*mean
    K8*mean K9*mean K10*mean K11*mean K12*mean K13*mean
    K14*mean K15*mean K16*mean; /*= Faktorscores auf Behörden-
    ebene für die einzelnen Kennzahlen*/
  weight DG_Gesamt; /*Designgewichtung*/
run;
```

[.....]

```
/*Berechnung der Wichtigkeit pro Kennzahl (WK1-WK16) pro Behörde pro
Schicht (Detailbefragung) */
```

```
proc tabulate data= E4&Daten._Kennzahlen_detail
  out= w_Amt_Schicht (drop= _type_ _page_ _table_);
  var wK1-wK16/*Interviewwerte der einzelnen Kennzahlen*/;
  class q22 /*LL*/ q32 /*Schicht*/ q23 /*Behörde*/ / missing;
  table q22*q32*q23,
    N wK1*mean wK2*mean wK3*mean wK4*mean wK5*mean wK6*mean
    wK7*mean wK8*mean wK9*mean wK10*mean wK11*mean wK12*mean
```

¹ Als „Score“ wird ein über eine Mittelwertbildung aggregiertes Einzelergebnis entlang der verschiedenen Auswertungsebenen bezeichnet.

```
wK13*mean wK14*mean wK15*mean wK16*mean; /*= Wichtigkeit
auf Behördenebene für die einzelnen Kennzahlen*/
weight DG_Gesamt; /*Designgewichtung*/
run;

/*Die jeweiligen durchschnittlichen Zufriedenheitsangaben aus der
Befragung werden mit der ebenfalls durchschnittlichen jeweiligen
Wichtigkeit gewichtet. Erst jetzt liegt pro Kennzahl ein über die
durchschnittliche Wichtigkeit gewichteter Zufriedenheitsscore vor*/

data E4&Daten._Score_Amt_Schicht;
merge K_Amt_Schicht_detail w_Amt_Schicht;
by schluesselE4;
wK99 = sum(of Mean_Amt_wK1-Mean_Amt_wK16);
run;

data E4&Daten._Score_Amt_Schicht;
set E4&Daten._Score_Amt_Schicht;

array Zufriedenheit {16} Mean_Amt_K1-Mean_Amt_K16;
array Wichtigkeit {16} Mean_Amt_wK1-Mean_Amt_wK16;
array ZScore_K {16};

do i= 1 to dim(ZScore_K);
    ZScore_K{i} = sum (Zufriedenheit{i} * Wichtigkeit{i});
end;
Behoerdenscore = sum(of ZScore_K1-ZScore_K16) / wK99;
run;
```

Um eine Zufriedenheit der nächsthöheren Gliederungsebene Schicht – den **Schichtscore** – zu berechnen, wird die Summe der Behördenscores für die jeweilige Schicht gewichtet mit der Behördenkontakinzidenz gebildet.

/*Berechnung der (über die Behördenkontakinzidenz) gewichteten Zufriedenheit pro Schicht pro Lebenslage */

```
data Berechnung_Schichtscore;
set ullm17.E4&Daten._Ergebnis_Amt_Schicht;
gew_Score_Amt = Behoerdenscore * &BehoerdenGW.;
run;

proc summary data=Berechnung_Schichtscore sum;
var gew_Score_Amt N_Amt_Sum;
class q32;
output out= Ausgabe_Score_Schicht
sum= Schichtscore N_Amt_Schicht;
run;
```

Analog wird bei der Aggregation auf die nächste Gliederungsebene – der Berechnung des **Lebenslagenscores** – vorgegangen. Der Lebenslagenscore berechnet sich aus der Summe der Schichtscores für die jeweilige Schicht gewichtet mit der Schichtinzidenz.

```

/*Berechnung der (über die Schichtinzidenz) gewichteten Zufrieden-
heit pro Lebenslage */

data E2&Daten._Schicht_Inzdaten;
merge E2&Daten._Schicht_Inzdaten ullm17.Schichtinzidenz_S ;
    by q32;
    gew_Score_Schicht = Score_Schicht * &SchichtGW.;
run;
proc summary data=E2&Daten._Schicht_Inzdaten sum;
    var gew_Score_Schicht N_Amt_Schicht;
    class q22;
    output    out= work.M1&Daten._Ausgabe_Score_LL
              sum= des Lebenslagenscores N_Amt_LL;
run;

```

Die durchschnittliche „Zufriedenheit mit der öffentlichen Verwaltung“ berechnet sich schließlich aus dem ungewichteten Durchschnitt² aller Lebenslagenscores.

Für diese vertikale Basis-Auswertung im sogenannten Modul 1 existiert ein spezielles Makro (%macro MODUL1(Daten)), welches alle durchschnittlichen gewichteten Zufriedenheiten auf den verschiedenen Ebenen – wo notwendig – unter der Verwendung der „weight“-Anweisung berechnet. Das Makro stellt sicher, dass alle Ergebnisse – sei es aus Bürger- oder Unternehmensdaten oder von bestimmten Teilmengen auf ein und dieselbe Weise erstellt werden.

Varianten der vertikalen Auswertung (auch als Module 2 und 3 bezeichnet) entstehen dadurch, dass Schichten zu sogenannten Grobschichten zusammengefasst werden, eine Ebene „eingeklappt“ bzw. über die Lebenslagen hinweg „eingezogen“ wird. Letztere Auswertung hat als Ziel die direkte Darstellung der Behördenzufriedenheiten unabhängig von den Schichten bzw. Lebenslagen.

Für diese beiden Varianten existieren ebenfalls spezielle Makros (%macro Behoerden_LL(Daten,LL) und %macro MODUL3(Daten)), die wie schon das Makro %macro MODUL1 dieselbe Auswertung für alle Daten gewährleistet. Beide Makros bauen auf der Basis des Behördenscores für alle Schichten aller Lebenslagen auf.

Über die Behördenkontakt- und Schichtinzidenz wird in Modul 2 über das Makro „Behoerden_LL“ ein Behördengewicht der einzelnen Behörde in der Lebenslage berechnet. Mit diesem Behördengewicht gewichtet wird aus den Behördenscores für alle Schichten aller Lebenslagen, die aus Modul 1 zur Verfügung stehen, der Behördenscore für die Lebenslage errechnet, diesmal ohne die Schichtebene darzustellen.

² Hintergrund der ungewichteten Durchschnittsberechnung hier ist der Gedanke, kein Lebenslagenranking zu generieren.

```
data E2&Daten._LL_Schicht_Amt (compress=yes);
set E2&Daten._LL_Schicht_Amt;
    BS_Sweight = &BehoerdenGW. * &SchichtGW.;
run;

proc summary data=E2&Daten._LL_Schicht_Amt sum;
    var BS_Sweight;
    class q22 q23;
    output    out= BS_Sweightsum
              sum= BS_Sweightsum;
run;

data LLBS_Sweightsum (drop=_TYPE_ _FREQ_ q23 BS_Sweightsum);
set BS_Sweightsum;
    if q22 ne . and q23 = .;
    LLBS_Sweightsum=BS_Sweightsum;
run;

data BS_Sweightsum (drop=_TYPE_ _FREQ_);
set BS_Sweightsum;
    if q22 ne . and q23 ne .;
run;

[...Sortierung und Merge von E2&Daten._LL_Schicht_Amt und
BS_Sweightsum ...]
/* Gewichtetes Mittel der Bördenscores pro Lebenslage über die
Schichten hinweg*/

proc means data=E2&Daten._LL_Schicht_Amt mean noprint;
    var Behoerdenscore;
    class q22 q23;
    weight Amtsgewicht;
    output    out= Ausgabe_Score
              mean= Modul2_Score_Amt;
run;
```

Analog des Makros „%macro MODUL1“ ist auch in Makro „%macro MODUL3“ die Schicht die nächst höhere Ebene. Der Unterschied besteht lediglich in der Zusammenfassung mehrere Schichten zu sogenannten Grobschichten. Die Schichtscores, die in Modul 1 berechnet wurden, werden zu Grobschichtscores zusammenfasst – gewichtet mit der jeweiligen Schichtinzidenz. Aus den Schichtinzidenzen werden Grobschichtinzidenzen berechnet, die die Gewichtung zum Lebenslagenscore ermöglicht.

Die jeweiligen Lebenslagenscores – egal ob aus Modul 1, 2 oder 3 berechnet – sind immer gleich. D. h. theoretisch müssten sie aus Modul 2 und 3 gar nicht berechnet werden. Sie werden dennoch ausgegeben; als Qualitätskontrolle und Plausibilitätskriterium für die Ergebnisse. Stimmen die Lebenslagenscores überein, sind die Ergebnisse über alle Ebenen und Module hin konsistent.

Die **horizontale Auswertung** generiert Ergebnisse zu den Zufriedenheiten und Wichtigkeiten für die 16 Kennzahlen. Hierzu werden zunächst die individuellen Zufriedenheiten und Wichtigkeiten je Interview, Kennzahl, Behörde und Schicht mit dem

individuellen Designgewicht gewichtet und die durchschnittliche Zufriedenheit und Wichtigkeit je Kennzahl pro Behörde berechnet.

Auch hierfür dient die aus Modul 1 berechnete durchschnittliche Zufriedenheit für jede der 16 Kennzahlen und für jede Behörde jeder Lebenslage als Ausgangsbasis.

*/*Berechnung der (über die Behoerdenkontaktinzidenz_bs) gewichteten Kennzahlen pro Schicht pro Lebenslage*/*

```
proc tabulate data= M5E4&Daten._KennZ_Amt_Schicht format=commax10.4
  out= Ausgabe_Kennzahlen_Schicht (drop= _type_ _page_ _table_);

  var   Mean_Amt_K1-Mean_Amt_K16   Mean_Amt_wK1-Mean_Amt_wK16;
  /*= Faktorscores auf Behördenebene für die einzelnen Kenn-
  zahlen, vgl. S. 10*/
  class q22 /*LL*/ q32 /*Schicht*/ / missing;
  table q22*q32,
    Mean_Amt_K1*mean      Mean_Amt_K2*mean      Mean_Amt_K3*mean
    Mean_Amt_K4*mean      Mean_Amt_K5*mean      Mean_Amt_K6*mean
    Mean_Amt_K7*mean      Mean_Amt_K8*mean      Mean_Amt_K9*mean
    Mean_Amt_K10*mean     Mean_Amt_K11*mean     Mean_Amt_K12*mean
    Mean_Amt_K13*mean     Mean_Amt_K14*mean     Mean_Amt_K15*mean
    Mean_Amt_K16*mean;
  weight Detailbehoerdengewicht;
run;
```

Im Anschluss erfolgt eine (mit der Schichtinzidenz) gewichtete Aggregation zur Lebenslage, so dass für jede Lebenslage 16 Kennzahl-Zufriedenheiten und 16 Kennzahl-Wichtigkeiten vorliegen. Die Werte der flachen Behörden können nicht mit einbezogen werden, da für sie nur eine Gesamtbewertung vorliegt, aber nicht die Zufriedenheit bzgl. der 16 Kennzahlen erfragt wurde. Selbstverständlich ist auch dieses Modul als Makro (%macro Faktoren(Daten)) aufgebaut, um die Konsistenz der Ergebniserstellung zu gewährleisten.

6 Teilmengen- und Streuungsauswertung

Die Teilmengen- und Streuungsauswertung muss in das System der horizontalen und vertikalen Basisauswertung eingepasst werden. Zunächst werden zahlreiche, für die interessierenden Teilmengen, individuelle Selektionen programmiert. Interessiert man sich beispielsweise für das Produzierende Gewerbe, werden die Interviews der Unternehmen mit den entsprechenden Wirtschaftszweigen gefiltert. Interessiert man sich für das Viertel mit der geringsten Zufriedenheit (Quartilsauswertung) werden die 25% Unzufriedensten pro Lebenslage, Schicht oder Behörde gesucht. Für jede abgeschlossene Filterung muss anschließend die oben beschriebene Basisauswertung laufen.

Die Herausforderung in Bezug auf die Teilmengen- und Streuungsauswertung als Ausgangspunkt für Hinweise über mögliche Vereinfachungs- und Optimierungspotentiale ist die Masse an Ergebnissen, die produziert, qualitätsgesichert und bewertet werden

muss, um im Vergleich die Bereiche zu identifizieren, deren weiterführende Analyse lohnen könnte.

In einem fiktiven Lebenslagen-Modell mit insg. 12 Lebenslagen, 20 Schichten und 25 Behörden erhält man für eine vertikale Basisauswertung ohne Varianten/ Module 58 aggregierte Einzelergebnisse plus mindestens 32 aggregierte Einzelergebnisse aus der horizontalen Basisauswertung (insgesamt 90 Ergebnisse). Alleine die vertikale Auswertung von fünf Beschäftigtengrößenklassen und deren Streuung ohne Varianten/ Module potenziert sich auf 15 Teilmengenfilterungen, 15 Durchläufe der Basisauswertung und 1.350 Einzelergebnisse. Spätestens hier ist ein effektives Management der Daten unumgänglich, damit Laufzeiten in Grenzen gehalten werden.

7 Grafische Darstellung

Die zuvor erzeugten SAS-Ergebnisse werden größtenteils in Form von Grafiken in einem Ergebnisbericht oder auf der Internetseite ‚www.amtlich-einfach.de‘ veröffentlicht. Diese grafische Darstellung der Ergebnisse erlaubt eine bessere Interpretation der Daten bzw. einen einfacheren Vergleich der Befragungswellen 2015 und 2017. Da diese Vielzahl von Grafiken immer wieder für alle Lebenslagen (22 bei Bürger- und 10 bei Unternehmensbefragung) erneut erstellt werden müssen (während der Erhebungsphase oder auch für zukünftige Folgebefragungen), sind hierfür mehrere SAS-Makros programmiert worden, um schnell und effizient unterschiedliche Auswertungsergebnisse – unter Verwendung gleicher Diagrammtypen – standardisiert grafisch darstellen zu können.

Da die ausgewerteten Ergebnisse für alle Lebenslagen oftmals in einer Gesamttabelle vorliegen, müssen für jede Lebenslage die entsprechenden Daten herausgefiltert werden, damit diese in den nachfolgenden Makros weiterverarbeitet werden können. Hierzu werden die ‚%do‘-, ‚%let‘- und ‚%scan‘-Anweisungen verwendet.

```
%macro filter_ll;
%do y=1 %to 22; /*22 Lebenslagen bei der Bürgerbefragung*/
  %let ll =%scan(11 12 13 14 15 16 17 18 22 23 24 25 26 27 28 29
                30 31 32 33 50 51, &y.);
  ...
%end;
%mend;
```

Die Inhalte bzw. die Funktion der Diagramm-Makros werden nachfolgend in aller Kürze skizziert:

Allgemeingültige Makroparameter für alle verwendeten Diagramm-Makros:

Seit der ersten Befragung (2015) werden die Bürgerergebnisse in blauen und die Unternehmensergebnisse in grünen Grafiken ausgewiesen. Damit das Makro für beide Befragungen genutzt werden kann, wird das Farbschema mit Angabe des Speicherorts (wird zuvor in den Templates definiert) als Makroparameter angegeben.

Im finalen Ergebnisbericht kann die Diagrammbreite variieren, wenn auf einigen Veröffentlichungsseiten noch eine Marginalspalte eingefügt wird. Deshalb lässt sich auch die benötigte Breite des jeweiligen Diagramms über die verschiedenen Makros einstellen.

In einigen Grafiken werden im zukünftigen Veröffentlichungsbericht die Ergebnisse aus aufeinanderfolgenden Befragungen (z. B. aus den Jahren 2015 und 2017) miteinander verglichen. Um für den Vergleich z. B. das bestehende Balkendiagramm-Makro verwenden zu können, besteht auch hier anhand eines Makroparameters die Möglichkeit, dass die Vergleichsdaten aus der Vorbefragung (z. B. aus dem Jahr 2015) – sofern gewünscht – hinzugespielt werden können.

Weiterhin wird jede SAS-Datendatei, die für eine Grafik verwendet wird, als TXT-Datei exportiert.

Die Daten in den TXT-Dateien werden mit anderen Softwareprogrammen weiterverarbeitet, um diese auf der Internetseite ‚www.amtlich-einfach.de‘ in Form von Grafiken (die dem Hausstandard entsprechen) oder Tabellen darzustellen.

Makros für Balken- und Streudiagramme:

Die Ergebnisse der Zufriedenheitsbefragung werden größtenteils in Form von horizontalen Balkendiagrammen dargestellt.

Da die Anzahl der Balken oder Linien (im Streudiagramm) für verschiedene Auswertungen variieren können, die Balken- und Streudiagramme aber immer das gleiche Layout aufweisen sollen, wird in den Grafikmakros die Anzahl der Balken bzw. Linien anhand der jeweiligen möglichen Merkmalsausprägungen berechnet. Über die Anzahl der Balken bzw. Linien lässt sich der Abstand der Kategorienbeschriftungen zum oberen und unteren Rand des Diagrammes festlegen sowie mit einem individuell bestimmten Balkenabstand die individuelle Länge der Y-Achse ermitteln, darüber hinaus wiederum die individuelle Papierhöhe der Grafik.

```
/*Balkenanzahl ermitteln*/
proc sort data=&ein_daten. out=anz;
    by anzahl;
run;

data anz_balken (keep=anzahl);
set anz;
if last.anzahl;
by anzahl;
call symput('order_count', anzahl);
run;
```

```
/*Abstand der Kategorienbeschriftungen zum oberen und unteren Rand
des Diagrammes */
```

```
%let offset = %sysevalf(0.65/&order_count.);

/* Länge der Y-Achse berechnet sich aus der Anzahl der Balken*/
%global graph_height;
%let length_y = %sysevalf(&order_count /2.3);
%put length_y = &length_y;

/* Papierhöhe berechnet sich aus der Länge der Y-Achse*/
data _null_;
  length graph_height 4;
  graph_height = &length_y +2.1;
  graph_height = round(graph_height, .01);
  call symputx("graph_height",graph_height);
run;
%put graph_height = &graph_height;
```

Weiterhin wird in dem Balken und Streudiagramm die individuelle Sortierung der Daten gesteuert. In einigen Ausgaben sollen die Balken bzw. Streulinien nach den absteigenden Zufriedenheitswerten dargestellt werden, in anderen nach vorgegebenen Größenklassen bzw. Kategorien mit vorbestimmter Reihenfolge (s. nachfolgendes Makro).

```
%macro Balken_2017 (
ein_daten      =      , /*Datensatz */
yvar           =      , /*Variable mit den Kategorien-Bezeichnungen */
var1           = na   , /*Möglichkeit, Variable mit Mittelwerten
                        aus 2015 anzugeben, die dann über Diamanten
                        dargestellt wird */
xvar2          =      , /*Variable mit Mittelwerten aus 2017 für
                        die Balken*/
refline1       = na   , /*Referenzlinie bei Gesamtmittel */
refline1label1 =      , /*Label zu Referenzlinie 1, muss in "" stehen */
labelpos       =start, /*Datalabels an Anfang (labelpos=start) oder
                        Ende (labelpos=ende) der Balken stellen */
datalabel      =na    , /*Variable mit Datalabels für die Balken */
sort           =data, /*standardmäßig absteigend, alternativ
                        sort=fest*/
sortvar        =      , /*nur bei fester Sortierung notwendig: Hier die
                        Variable angeben nach der Sortiert werden
                        soll*/
branche        =0     , /*bei branche=1 wird das zweistufige Farbschema
                        für Untergruppen angewandt */
farbschema     =      , /*nichts für Unternehmen (grün) und _b für
                        Bürger eingeben (blau) */
text           =1     , /*Wenn hier 0 eingegeben wird, dann wird keine
                        Textdatei mit den Daten erstellt. */
breite         =intro, /*breite = intro standardmäßig Grafikbreite
                        von 18.2 cm (f. Intro); breite = steck
                        Grafikbreite von 13.5 cm (f. Steckbriefe) */
abb_nr         =      , /*Bildname zum abspeichern (pdf und png) */);
```

Soll das Balkendiagramm eine Ober- und Untergruppierung in Form von unterschiedlich gefärbten Balken darstellen (s. Abbildung 4 Zufriedenheit nach Branchen), so lässt sich über das o. gezeigte Makro mit dem Parameter ‚branche‘ ein weiteres spezielles Farbschema ein- und ausschalten.

Nachfolgend einige Beispiele aus der Unternehmensbefragung 2015:




Abbildung 2: Zufriedenheit mit den behördlichen Dienstleistungen im Rahmen ausgewählter Lebenslagen




Abbildung 3: Streuung und Zufriedenheit im Rahmen ausgewählter Lebenslagen (Streudiagramm)



Abbildung 4: Zufriedenheit der Unternehmen mit den behördlichen Dienstleistungen nach Branchenzugehörigkeit

Hinzu kommt noch die Problematik, dass die Y-Achsenbeschriftungen in den einzelnen Grafiken unterschiedlich lang sind und der jeweilige Null-Punkt der X-Achse im Veröffentlichungsbericht nicht immer an der gleichen Position ist. Dies kann die Lesbarkeit der Grafiken pro Seite erheblich beeinträchtigen. Somit wird für jede Seite des Veröffentlichungsberichts die maximale Y-Achsenbeschriftung sowie die daraus resultierende Position des Null-Punktes ermittelt und über die ‚Pad‘-Options die komplette Grafik entsprechend nach links verschoben. Dadurch verkleinert sich allerdings der Grafikbereich, die sog. Wall Area, entsprechend in der Breite.

```
%if &ein_daten. eq erwerbsstatus %then %do;
  %let links = 14;
%end;

proc sgplot data=&ein_daten. noborder nowall noautolegend pad=(top=0
right=0 bottom=0 left=&links.);
```

Achsenkreuze

Bei dem Achsenkreuz (s. Abb. 5 u. 6) handelt es sich um einen Scatterplot, bei dem nachträglich mithilfe einer anderen Software eine Verbindungslinie zwischen Punkt und Text eingefügt wird, wenn der Text aus Platzmangel nicht unmittelbar neben dem Punkt dargestellt werden kann.

Da im Bürger- sowie auch im Unternehmens-Veröffentlichungsbericht mehrere Achsenkreuze mit unterschiedlichen Inhalten abgebildet werden, kann die Achsenbeschriftung ebenfalls über Makroparameter gesteuert werden.

```
%macro achsenkreuz(
ein_daten =      ,/*Datensatz */
type           =ZXW ,/*Zufriedenheit X Wichtigkeit. Alternativ
                    type=ZXH mit Häufigkeit. Oder ZXK mit
                    Kompliziertheit (wichtig f. Beschriftungen)*/
xvar           =      ,/*Variable für die x-Achse (bei type=ZXW z.B.
                    Wichtigkeit, bei type=ZXH z.B. Häufigkeit) */
yvar           =      ,/*Variable für die y-Achse (i.d.R. Zufriedenheit) */
skala_fest_xmin =na,/*Hier besteht die Möglichkeit, die Minima
                    und Maxima der Skalen selber anzugeben. */
skala_fest_xmax =na,/*...Falls hier nichts angegeben wird, werden
                    Minima und Maxima automatisch anhand */
skala_fest_ymin =na,/*...der Daten bestimmt, sodass die Punkte
                    breit über die beiden Achsen streuen. */
skala_fest_ymax =na,/*...Nur durch 0.5 teilbare Werte hier möglich
                    (Z.B. 2.5, 3.0 oder 3.5)!! */
labelvar       =      , /*Variable mit Labels für die Datenpunkte */
abb_nr         =      /*Bildname zum abspeichern (pdf und png) */ );
```



Abbildung 5: Allgemeine Zufriedenheit und Häufigkeit der Lebenslagen




Abbildung 6: Zufriedenheit und Wichtigkeit der Faktoren über alle Lebenslagen

Kreisdiagramme mit SAS Graph

Die Kreisdiagramme sind die einzigen Diagramme, die mit SAS Graph erstellt werden. Über das Kreisdiagramm-Makro kann man die jeweilige Darstellungsform – ob einfache Ringe (s. Abb. 7) oder Doppelringe (s. Abb. 8) – steuern.

```
%macro Ringe_kontakt(
ein_daten = , /*Datensatz */
abb_nr = , /*Bildname zum abspeichern (pdf und png) */
farbschema = , /*nichts für Unternehmen (grün) und _b für Bürger
eingeben (blau) */
text =1 , /*Wenn hier 0 eingegeben wird, dann wird keine
Textdatei mit den Daten erstellt. */
art =einfach /*einfach (Einfachring) oder ineinander
(Doppelring)*/
);
```

Leider gibt es bei der Erstellung der Kreisdiagramme mit der SAS-Version 9.4 Probleme mit dem Speicherort sowie der Darstellung als PNG- und PDF-Datei. Die Kreisgrafiken werden direkt unter dem SAS-HOMES-Verzeichnis mit dem Namen ‚sas-graph.png‘ sowie im angegebenen Speicherort mit dem Namen ‚ganno.png‘ doppelt abgelegt. Die zusätzlich angelegte PDF- sowie auch PNG-Datei im angegebenen Speicherort wird zudem in einer qualitativ minderwertigen Auflösung dargestellt, so dass man zusätzlich die PNG-Datei unter dem SAS-HOMES-Verzeichnis manuell in eine PDF-Datei konvertieren muss. Bei der 2015er Befragung wurde anfänglich im Hause noch die SAS-Version 7.2 verwendet. Mit dieser älteren SAS-Version konnten die Grafiken mit dem Kreisdiagramm-Makro ohne Probleme erstellt werden.



Abbildung 7: Art der Kommunikation mit Behörden in 2015 über alle Lebenslagen Telefon, Brief, Fax, online, persönlicher Besuch bei Behörde, persönlicher Besuch durch Behörde (Mehrfachnennungen möglich, Angaben in Prozent der Befragten)



Abbildung 8: Art der Kommunikation mit Behörden in 2017 über alle Lebenslagen Telefon, Brief, online, persönlicher Besuch bei Behörde (Mehrfachnennungen möglich, Angaben in Prozent der Befragten; äußerer Ring: 2017, innerer Ring 2015) *(Fiktive Daten, da die 2017er Ergebnisse parallel zum Erscheinen dieses Artikels erst noch veröffentlicht werden.)*

Literatur

- [1] Hochgürtel, T. Lösch, T. (2013) Analysen mit der DRG-Statistik – Herausforderungen und Lösungsansätze. Beitrag zur 17. Konferenz der SAS-Anwender in Forschung und Entwicklung (KSFE) Ulm, 2013.
- [2] Kühnhenrich, D. und Michalik, S. (2016) Are Citizens and Businesses (Dis)satisfied with the Public Administration in Germany? Beitrag zur 34. IARIW General Conference Dresden, August, 2016.
- [3] Himmelsbach, E., Hornbach, C., Michalik, S. und Kühnhenrich, D. (2016) Methodische Grundlagen der Zufriedenheitsbefragungen zu behördlichen Dienstleistungen. In: WISTA Wirtschaft und Statistik, Wiesbaden, Ausgabe 4/2016, Seite 54 ff.
- [4] Schmidt, B., Kühnhenrich, D., Zipse, C., und Vorgrimler, D. (2015) Entlastungen spürbarer machen – Wie wird der Kontakt zur Verwaltung wahrgenommen? In: WISTA Wirtschaft und Statistik, Wiesbaden, Ausgabe 2/2015, Seite 56 ff.
- [5] Statistisches Bundesamt (2016a) Wahrnehmung von bürokratischen Belastungen durch Unternehmen in ausgewählten Situationen in Deutschland. Statement zur Pressekonferenz am 21. Januar 2016 in Berlin, Online verfügbar unter

- https://www.destatis.de/DE/PresseService/Presse/Pressekonferenzen/2016/zufriedenheitsbefragung/Statement_Sarreither_unternehmenssituationen.pdf?__blob=publicationFile (abgerufen am 19.02.2018)
- [6] Statistisches Bundesamt (2016b) Wahrnehmung von bürokratischen Belastungen durch Unternehmen in ausgewählten Situationen in Deutschland– Ausgewählte Ergebnisse der Zufriedenheitsbefragung 2015. Publikation zur Pressekonferenz am 21. Januar 2016 in Berlin, Online verfügbar unter https://www.destatis.de/DE/PresseService/Presse/Pressekonferenzen/2016/zufriedenheitsbefragung/Pressebrochuere_unternehmenssituationen.pdf?__blob=publicationFile (abgerufen am 19.02.2018)
- [7] Statistisches Bundesamt (2016c) 61 % der Unternehmen in Deutschland legen Wert auf E-Government. Pressemitteilung zur Pressekonferenz am 21. Januar 2016 in Berlin, Online verfügbar unter https://www.destatis.de/DE/PresseService/Presse/Pressekonferenzen/2016/zufriedenheitsbefragung/pm_unternehmenssituationen_PDF.pdf?__blob=publicationFile (abgerufen am 19.02.2018)
- [8] Statistisches Bundesamt (2015a) Zufriedenheit der Bürgerinnen und Bürger in Deutschland mit behördlichen Dienstleistungen – Ausgewählte Ergebnisse der Zufriedenheitsbefragung 2015. Statement zur Pressekonferenz am 19. August 2015 in Berlin, Online verfügbar unter https://www.destatis.de/DE/PresseService/Presse/Pressekonferenzen/2015/zufriedenheitsbefragung/Statement_Egeler_zufriedenheitsbefragung.pdf?__blob=publicationFile (abgerufen am 19.02.2018)
- [9] Statistisches Bundesamt (2015b) Zufriedenheit der Bürgerinnen und Bürger in Deutschland mit behördlichen Dienstleistungen – Ausgewählte Ergebnisse der Zufriedenheitsbefragung 2015. Begleitheft zur Pressekonferenz am 19. August 2015 in Berlin, Online verfügbar unter https://www.destatis.de/DE/PresseService/Presse/Pressekonferenzen/2015/zufriedenheitsbefragung/Pressebrochuere_zufriedenheitsbefragung.pdf?__blob=publicationFile (abgerufen am 19.02.2018)
- [10] Statistisches Bundesamt (2015c) Bürger sind mit ihrer öffentlichen Verwaltung überwiegend zufrieden. Pressemitteilung zur Pressekonferenz am 19. August 2015 in Berlin, Online verfügbar unter https://www.destatis.de/DE/PresseService/Presse/Pressekonferenzen/2015/zufriedenheitsbefragung/pm_zufriedenheitsbefragung_PDF.pdf?__blob=publicationFile (abgerufen am 19.02.2018)